
On souhaite concevoir deux applications manipulant des échantillons de nombres réels et utilisant 
leur moyenne et leur variance. Pour éviter de recopier deux fois les mêmes traitements, on utilisera 
un module (ensemble de données et de procédures ou fonctions) partagé par les deux applications. 
 
 
Algorithmique (du module) 
 
1) Déclarer une constante NMAX dont la valeur sera le nombre maximum de valeurs dans un 

échantillon. Par exemple, fixer la valeur de NMAX à 25. 

2) Déclarer un type enregistrement Echantillon avec deux champs : un tableau de réels nommé x 
et un entier nommé n dont la valeur sera la taille véritable de l’échantillon (n ≤ NMAX). 

3) Écrire une procédure de calcul de moyenne et de variance. Paramètre en entrée (par valeur) : 
une variable de type Echantillon. Paramètres en sortie (par adresse) : moyenne et variance. 
Rappel des formules : 

Const NMAX = 100 
 
Type Echantillon = Enregistrement 
  n : entier 
  x : tableau(NMAX) de réels 
Fin 
 
Procédure MoyVar(ech : Echantillon, var m : réel, var v : réel) 
 
Var i : entier 
    somme, somme2 : réel 
 
Début 
 somme = 0 
 somme2 = 0 
 Pour i := 1 à ech.n faire 
  somme := somme + ech.x(i) 
  somme2 := somme2 + ech.x(i) ^ 2 
 Fin pour 
 m = somme / ech.n 
 v = √((ech.n * somme2 - somme ^ 2) / ech.n ^ 2) 
Fin 

Visual Basic – Application 1 : Population 
 
Concevoir une application ayant l’apparence suivante. 
 

 
 
Les deux zones de texte permettent de saisir des noms de pays et leur population (en millions). Un 
premier bouton de commande permet d’ajouter ces données dans une liste pour affichage à l’écran 
et de stocker les populations dans un tableau pour traitement ultérieur. Le second bouton de 
commande déclenche les calculs effectués à partir des valeurs contenues dans le tableau. 
 
1) Ajouter un module au projet (menu Projet/Ajouter un module/Nouveau). 

2) Au sein du module, déclarer un type Echantillon personnalisé de la façon suivante 
(enregistrement). 

 
' Taille maxi des tableaux 
 
Public Const NMAX = 100 
 
' Type échantillon avec taille réelle et tableau de valeurs 
 
Public Type Echantillon 
    n As Integer 
    x(NMAX) As Single 
End Type 

 
3) Au sein du module, saisir la procédure de calcul de moyenne et de variance. Les paramètres par 

adresse doivent être précédés du mot-clé ByRef. 
4) Enregistrer le module. Revenir à l’application (la feuille) grâce au gestionnaire de projets 

(CTRL+R). 

5) Dessiner l’interface de l’application. 

6) Déclarer une variable globale de type Echantillon et écrire les procédures _click attachées aux 
boutons de commande de la feuille. Appeler dans la partie calcul la procédure définie dans le 
module. 

Code du module : 
 
' Taille maxi des tableaux 
 
Public Const NMAX = 100 
 
' Type échantillon avec taille réelle et valeurs 
 
Public Type Echantillon 
    n As Integer 
    x(NMAX) As Single 
End Type 
 
' Moyenne / Variance 
 
Public Sub MoyVar(ech As Echantillon, ByRef m As Single, ByRef v As Single) 
    ' Déclaration des variables locales 
    Dim i As Integer 
    Dim somme As Single, somme2 As Single 
    ' Initialisations 
    somme = 0 
    somme2 = 0 
    ' Cumuls 
    For i = 1 To ech.n 
        somme = somme + ech.x(i) 
        somme2 = somme2 + ech.x(i) ^ 2 
    Next 
    ' Calculs 
    m = somme / ech.n 
    v = Sqr((ech.n * somme2 - somme ^ 2) / ech.n ^ 2) 
End Sub 

 
Code de la feuille : 
 
Dim don As Echantillon 
 
Private Sub Command1_Click() ' Ajout 
    If Pays.Text <> "" And Population.Text <> "" And Liste.ListCount < NMAX Then 
        ' Ajout dans la liste pour affichage 
        Liste.AddItem (Pays.Text & " : " & Population.Text) 
        ' Ajout dans le tableau 
        don.n = don.n + 1 
        don.x(don.n) = CSng(Population.Text) 
        ' Effacement des zones de saisie 
        Pays.Text = "" 
        Population.Text = "" 
    Else 
        Beep 
    End If 
End Sub 
 
Private Sub Command2_Click() ' Calcul 
    ' Déclarations 
    Dim moy As Single 
    Dim vari As Single 
    ' Appel à la procédure du module 
    Call MoyVar(don, moy, vari) 
    ' Affichage du résultat 
    Moyenne.Caption = "Moyenne = " & CStr(moy) 
    Variance.Caption = "Variance = " & CStr(vari) 
End Sub 

Visual Basic – Application 2 : Stock 
 
Concevoir une application ayant l’apparence suivante. 
 

 
 
La zone de texte sert à saisir la valeur de lots et le bouton de commande sert à ajouter les lots au 
stock existant. À chaque fois qu’un lot est ajouté, on souhaite connaître la valeur totale du stock et 
sa valeur moyenne par lot. 
 
1) Ajouter un module au projet (menu Projet/Ajouter un module/Existant — sélectionner celui qui 

vient d’être créé pour l’application 1). 

2) Dessiner l’interface de l’application. 

3) Déclarer une variable globale de type Echantillon et écrire la procédure _click attachée au 
bouton de commande de la feuille. Réutiliser la procédure de calcul de moyenne/variance 
définie dans le module. 

 
Code de la feuille : 
 
' Variables globales 
Dim stock As Echantillon 
Dim total As Single 
 
Private Sub Ajouter_Click() 
    ' Déclarations 
    Dim x As Single 
    Dim m As Single 
    Dim v As Single 
    If stock.taille < NMAX Then 
        ' Conversion valeur lot 
        x = CSng(Val_lot.Text) 
        ' Calcul du total 
        total = total + x 
        ' Insertion dans le tableau stock 
        stock.n = stock.n + 1 
        stock.x(stock.n) = x 
        ' Appel au calcul de moyenne 
        Call MoyVar(stock, m, v) 
        ' Affichage du résultat 
        Val_tot.Caption = "Valeur totale = " & Str(total) 
        Val_moy.Caption = "Valeur moyenne =" & Str(m) 
        ' Remise à zéro du champ de saisie 
        Val_lot.Text = "" 
    Else 
        Beep 
    End If 
End Sub 


